# $Id$
# Maintainer: Giovanni Scafora <giovanni@archlinux.org>
# Contributor: Tom Newsom <Jeepster@gmx.co.uk>

pkgname=strace
pkgver=4.18
pkgrel=1
pkgdesc='A diagnostic, debugging and instructional userspace tracer'
arch=(i686 x86_64)
url='http://sourceforge.net/projects/strace/'
license=(BSD)
depends=(perl libunwind)
source=(http://downloads.sourceforge.net/$pkgname/${pkgname}-${pkgver}.tar.xz{,.asc})
md5sums=('3579b3266bb096cebaefbe2cdb1a3a78'
         'SKIP')
validpgpkeys=('296D6F29A020808E8717A8842DB5BD89A340AEB7') # Dmitry V. Levin <ldv@altlinux.org>

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr \
    --with-libunwind
  make
}

check() {
  make -C $pkgname-$pkgver check
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
  install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
